
Лабораторная работа  1
ЗНАКОМСТВО С ИНТЕРФЕЙСОМ IDE AVR STUDIO

Краткие теоретические сведения

В поддержку своей архитектуры в 1997 г. Atmel выпустила AVR Studio - программный продукт для разработки приложений на основе AVR-микроконтроллеров. AVR Studio представляет собой интегрированную среду разработки IDE (Integrate Development Environment), объединяя в себе большое количество различных инструментов для написания и отладки программ. Продвигая на рынке новую продукцию, Atmel с самого начала попыталась сделать ее максимально открытой для потребителя. AVR Studio не является исключением. Последняя версия IDE, равно как и любая информация по AVR-микроконтроллерам, всегда свободно доступна на сайтах производителя.
Минимальный набор разработчика в IDE представлен фирменным ассемблером и симулятором. Однако AVR Studio легко интегрируется со многими программными средствами сторонних производителей. И на сегодняшний день, в частности, в нее включена поддержка постоянно развивающегося компилятора языка Си WinAVRоснованного на принципах GNU. Это очень мощный и к тому же бесплатный инструмент, что очень редко встречается у микроконтроллеров уровня AVR.
Ассемблер относится к языкам программирования низкого уровня. Его основой является множество команд, уникальное для каждого микропроцессора. Поэтому ассемблер является также и аппаратно-зависимым. Он может использоваться только совместно с архитектурой определенного типа. Каждое семейство микропроцессоров имеет свой собственный вариант этого языка.
Каждая инструкция ассемблера представляет собой символическое изображение соответствующей машинной команды со своим кодом операции (КОП). Команды ассемблера имеют удобочитаемый вид и названия, ассоциирующиеся с их действием. Так команда пересылки между двумя РОНами mov Rd,Rr является прототипом 16-разрядного кода операции 0010 11rd dddd rrrr. Битовые поля ddddd и rrrrr в нем определяют адреса регистров приемника и источника соответственно. Например, для пересылки регистра R5 в R22 необходимо записать на ассемблере mov R22,R5 или 0010 1101 0110 0101 на машинном языке. Разница в восприятии очевидна.
Каждую команду ассемблера можно логически разделить на две части: мнемонику и операнды.
Мнемоника   Операнд1   Операнд 2
 mov                 Rd,              Rr

Мнемоника является обязательной частью команды и определяет ее функциональное назначение. Операнды представляют собой параметры команды. В качестве операндов могут выступать числовые значения, адреса и смещения относительно адресов. Команды AVR, в зависимости от назначения, могут иметь до двух параметров.
Главная задача ассемблера – преобразование исходного текста пользовательской программы в машинный код, пригодный для записи в память программ микроконтроллера. Сам процесс преобразования называют компиляцией, а программу ассемблер – компилятором.
Конечным результатом работы компилятора является файл с исполняемым кодом. Если программа использует инициализированные данные, размещенные в энергонезависимой памяти, то в дополнении к этому, будет сгенерирован также файл для EEPROM. Информация, размещенная в этих файлах, используется программатором при программировании FLASH-память программ и/или EEPROM-память данных.
Существует большое количество различных форматов выходных файлов, но самый распространенный из них 16-тиричный Intel Hex Format. Файлы такого типа, как правило, имеют два различных расширения: .hex у файлов содержащих коды программ, и .epp у файлов, содержащих данные для записи EEPROM-памяти.
В соответствии со своей технологией работы, компилятор создает также объектный файл с расширением .obj. В нем размещена служебная информация для внутреннего пользования (относительные и абсолютные адреса операндов, место расположения объектов в пределах сегмента и т.д.). Объектный файл может быть необходим для различных отладочных средств. Программисту же никогда не приходится вмешиваться в его содержимое.
После сборки проекта могут быть сгенерированы также файлы, имеющие расширение .lst (файл листинга), .map и некоторые другие. В первом из них находится полный отчет о проделанной компилятором работе. Во втором приводится перечень всех символьных имен, встретившихся в программе, и их числовых значений.
Исходный текст программы на языке ассемблера может быть создан в любом текстовом редакторе и должен быть размещен в файлах с расширениями .asm, .inc или .txt. Удобнее всего для этих целей использовать встроенный в AVR Studio редактор.
Алфавит ассемблера состоит из набора символов ACSII. Однако в комментариях к программам допускается использовать любой удобный язык, основанный на наборе символов в кодировке UNICODE. Изначально ассемблер не чувствителен к регистру символов. Так, например, между тремя именами переменной TEMP, temp, Temp нет ни какой разницы.
Ассемблер может работать с числами, представленными в одной из четырех системах исчисления:
десятичной (примеры чисел: 10, 255), 
шестнадцатеричной (примеры чисел: 0x0A, $FF), 
двоичной (примеры чисел: 0b00001010, 0b11111111), 
восьмеричной (примеры чисел: 012, 0377).

По умолчанию применяется десятичная система, и любое число без префикса будет воспринято именно как десятичное. Необходимо обратить внимание также на то, что ассемблер AVR не поддерживает распространенную форму записи 16-тиричных чисел c постфиксом “H” (1AH, 0FFH).
В табл.1 приведен список операторов, доступных для использования в числовых выражениях. Операторы расположены в порядке убывания приоритета. Как и всегда, очередность следования арифметических и логических операций можно изменять, применяя круглые скобки.
Табл 1. Арифметические и логические операторы, использующие в числовых выражениях:
	Обозначение оператора
	Описание
	Пример использования

	-
	Унарный оператор. Изменяет знак числа, используя дополнительный код.
	ldi R16,-3   R16 <- 0xFD

	*
	Бинарный оператор. Производит умножение двух чисел.
	ldi R16,x1 * x2 ;R16 <- x1*x2 

	/
	Бинарный оператор. Производит деление двух чисел.
	ldi R16,x1 / x2 ;R16 <- x1/x2 

	%
	Бинарный оператор. Возвращает целочисленный остаток от деления двух чисел.
	ldi R16,x1 % x2 ;R16 <- {x1/x2} 

	!
	Унарный оператор сравнения. Возвращает 1 если число было равно нулю и 0 если число не было равно нулю.
	ldi R16,!x   R16 <- 1 если x = 0
;R16 <- 0 если x ≠ 0 

	~
	Унарный оператор. Производит инвертирование всех битов в числе.
	ldi R16,~0xF0   R16 <- 0x0F

	+
	Бинарный оператор. Производит сложение двух чисел.
	ldi R16,x1 + x2 ;R16 <- x1 + x2 

	-
	Бинарный оператор. Производит вычитание двух чисел.
	ldi R16,x1 - x2 ;R16 <- x1 - x2 

	<<
	Бинарный оператор. Производит сдвиг выражения стоящего слева от оператора на число разрядов, стоящих справа от оператора. Сдвиг происходит в левую сторону.  
	ldi R16,1<<5   R16 <- 0b00100000

	>>
	Бинарный оператор. Производит сдвиг выражения стоящего слева от оператора на число разрядов, стоящих справа от оператора. Сдвиг происходит в правую сторону. 
	ldi R16,0x50>>4 ;R16 <- 0b00000101

	<
	Бинарный оператор сравнения. Возвращает 1 если число, стоящее слева от оператора, меньше числа, стоящего справа. В противном случае возвращает 0.
	ldi R16,x1 < x2 ;R16 <- 1 если x1 < x2 
;R16 <- 0 если x1 >=x2 

	<=
	Бинарный оператор сравнения. Возвращает 1 если число, стоящее слева от оператора, меньше или равно числу, стоящему справа. В противном случае возвращает 0.
	ldi R16,x1 <= x2 ;R16 <- 1 если x1 <=x2 
;R16 <- 0 если x1 > x2 

	>
	Бинарный оператор сравнения. Возвращает 1 если число, стоящее слева от оператора, больше числа, стоящего справа. В противном случае возвращает 0.
	ldi R16,x1 > x2 ;R16 <- 1 если x1 > x2 
;R16 <- 0 если x1 <=x2 

	>=
	Бинарный оператор сравнения. Возвращает 1 если число, стоящее слева от оператора, больше или равно числу, стоящему справа. В противном случае возвращает 0.
	ldi R16,x1 >= x2 ;R16 <- 1 если x1 >=x2 
;R16 <- 0 если x1 < x2 

	==
	Бинарный оператор сравнения. Возвращает 1 если числа, стоящие по обе стороны от оператора, равны. В противном случае возвращает 0.
	ldi R16,x1 == x2 ;R16 <- 1 если x1 = x2 
;R16 <- 0 если x1 ≠ x2 

	!=
	Бинарный оператор сравнения. Возвращает 1 если числа, стоящие по обе стороны от оператора, не равны. В противном случае возвращает 0.
	ldi R16,x1 != x2 ;R16 <- 1 если x1 ≠ x2 
;R16 <- 0 если x1 = x2 

	&
	Бинарный оператор. Производит побитовую операцию “Логическое И” между двух чисел.
	ldi R16,x1 | x2 ;R16 <- x1|x2 

	^
	Бинарный оператор. Производит побитовую операцию “Исключающее ИЛИ” между двух чисел.
	ldi R16,x1 ^ x2 ;R16 <- x1^x2 

	|
	Бинарный оператор. Производит побитовую операцию “Логическое ИЛИ” между двух чисел.
	ldi R16,x1 & x2 ;R16 <- x1&x2 

	&&
	Бинарный оператор сравнения. Возвращает 1 если числа, стоящие по обе стороны от оператора, не равны нулю. В противном случае возвращает 0.
	ldi R16,x1 && x2 ;R16 <- 1 если x1 и x2 ≠ 0 
;R16 <- 0 если x1 или x2 = 0

	||
	Бинарный оператор сравнения. Возвращает 1 если хотя бы одно из чисел, стоящих по обе стороны от оператора, не равно нулю. В противном случае возвращает 0.
	ldi R16,x1 || x2 ;R16 <- 1 если x1 или x2 ≠ 0 
;R16 <- 0 если x1 и x2 = 0


Ассемблер содержит также ряд встроенных функций. Некоторые из них, имеющие наибольшее практическое значение, приведены в табл.2.

Табл 2. Некоторые встроенные в ассемблер функции:
	Обозначение функции
	Описание
	Пример использования

	low()
	Функция возвращает младший байт выражения.
	low(0x1234)  = 0x34

	high(), byte2()
	Функция возвращает второй байт выражения.
	high(0x1234) = byte2(0x1234) = 0x12

	byte3()
	Функция возвращает третий байт выражения.
	byte3(0xABCD1234) = 0xCD

	byte4()
	Функция возвращает четвертый байт выражения.
	byte4(0xABCD1234) = 0xAB

	lwrd()
	Функция возвращает младшее слово выражения.
	lwrd(0xABCD1234) = 0x1234

	hwrd()
	Функция возвращает второе слово выражения.
	hwrd(0xABCD1234) = 0xABCD

	abs()
	Функция возвращает абсолютное значение выражения.
	abs(-1000) = 1000


Операторы арифметических и логических операций, а также символы “;”, “.”, “(”, “)”, “[”, “]”, “{”, “}”, “?” - зарезервированы компилятором. Они не должны встречаться в обозначениях констант, инструкций и меток. Кроме этого, пользовательские имена, не могут начинаться с цифр и не должны совпадать с названиями регистров (R0…R31,X,Y,Z) и встроенными в ассемблер функциями.
Директивы ассемблера представляют собой команды управления компилятором. Объявление каждой из них должно начинается с точки. Практика показывает, что в любом ассемблере наиболее интенсивно используется только порядка 10…20 директив. Все остальные либо не являются обязательными, либо отвечают за управление, лишь незначительными свойствами компилятора. К “основным”, характерным и для ассемблеров других процессоров, относятся директивы .equ, .org, .def, .сseg, .dseg и т.д. Ну, а такие директивы, как .dq, .exit, .listmac  в реальных программах встречаются действительно очень редко. Ниже приведен перечень, описание и примеры использования директив фирменного ассемблера микроконтроллеров AVR.
Директива .include подставляет текстовый файл в то место программы, где происходит ее употребление. В дополнении к этому сам файл подстановки также может содержать директиву .include. Если файл расположен в директории проекта или в одной из служебных папок, то вместо полного пути, допускается указывать, лишь ссылку на его имя.
Директива .include
Синтаксис написания: 
.include "{путь к файлу}"
Пример использования:
	1
	.include  "m8def.inc" ;вставка стандартного заголовочного файла


Директива .exit указывает ассемблеру место окончания файла исходного текста. Все операторы, находящиеся после директивы, становятся невидимыми для компилятора. Если .exit встречается в подключаемом файле, то сборка проекта заканчивается строкой, где расположена директива .include. В случае отсутствия директивы .exit, конечной точкой сборки считается последняя строка исходного текста.
Директива .exit
Синтаксис написания:
.exit
Пример использования:
	1
	.exit ;конец файла 


Директивы .nolist и .list служат для управления файлом листинга, который обычно генерируется после сборки проекта. Первая из них запрещает, а другая, соответственно, разрешает вывод информации в файл. Директива .list отменяет действие .nolist и наоборот. 
Директивы .nolist, .list
Синтаксис написания:
.nolist, .list
Пример использования:
	1
2
3
	.nolist              ;запретить вывод текста файла “m8def.inc”  
.include "m8def.inc" ;в файл листинга программы
.list                ;продолжить вывод информации


Директива .equ присваивает символьному имени некоторое числовое значение. Символьное имя должно быть уникальным и не может быть изменено в процессе написания программы. Директива не может применяться для назначения символьных имен регистрам общего назначения.
Директива .equ
Синтаксис написания:
.equ {символьное имя} = {выражение}
Пример использования:
	1
2
	.equ DDRB  = 0x17    ;присвоение имени DDRB значения 0x17
.equ PORTB = DDRB + 1 ;присвоение имени PORTB значения 0x18


Директива .set производит то же самое действие, что и .equ. Но в отличии от последней, символьное имя может быть переопределено в любом месте программы.
Директива .set
Синтаксис написания:
.set {символьное имя} = {выражение}
Пример использования:
	1
2
3
	.set OFFSET = 0x100    ;присвоение имени OFFSET значения 0x100
.
.set OFFSET = OFFSET + 1 ;переопределение значения OFFSET


Директива .def присваивает символьное имя одному из регистров общего назначения. В дальнейшем ходе программы данное имя может быть отменено директивой .undef. 
Директивы .def, .undef
Синтаксис написания:
.def {символьное имя} = {регистр}
.undef {символьное имя} 
Пример использования:
	1
2
	.def temp = R16 ;присвоение регистру R16 имя temp
.undef temp    ;отмена дальнейшего использования имени temp


Директивы .db, .dw, .dd, .dq предназначены для резервирования памяти микроконтроллера под инициализированные данные. Все они могут применяться только в сегментах кода и EEPROM-памяти. Разница между этими директивами заключается в разрядности, представляемых данных. Директива .db резервирует байты, .dw – слова, .dd – двойные слова. В редких случаях может так же оказаться удобным использование директивы .dq, резервирующей 64-разрядные данные.
Директивы .db, .dw, .dd, .dq
Синтаксис написания:
{метка}: .db {8-разрядные данные}
{метка}: .dw {16-разрядные данные}
{метка}: .dd {32-разрядные данные} 
{метка}: .dq {64-разрядные данные} 
Пример использования:
	1
2
3
4
5
	label: 
 .db 0xFA, 250, -6, 0b11111010
 .dw 0xFADE, 64222, -1314, 0b1111101011011110 
 .dd 0xFADEEFCA, 4208914378, -86052918
 .dq 0xFADEEFCAEFBACDEF, 18077149609196178927, -521103510453211


Директива .byte резервирует память под неинициализированные данные в сегментах SRAM и EEPROM.
Директива .byte
Синтаксис написания:
{метка}: .byte {количество резервируемых данных} 
Пример использования:
	1
2
	.equ PAGESIZE = 0x20
buffer: . byte 2*PAGESIZE ;резервирование 64 байт в SRAM 


Директивы .dseg, .eseg, .cseg определяют начало сегментов данных, EEPROM-памяти и кода соответственно. В исходном файле каждый из сегментов может быть представлен только в одном экземпляре. В случае если все эти директивы отсутствуют в программе, компилятор по умолчанию считает, что все операторы расположены в секции кода.
Директивы .dseg, .eseg, .cseg
Синтаксис написания:
.dseg
.eseg
.cseg
Пример использования:
	1
2
3
4
5
6
7
8
9
10
11
	   .dseg         ;начало сегмента данных 
buffer: . byte 32 ;резервирование 32 байт под буфер в SRAM
 
   .cseg         ;начало сегмента кода
   rjmp  initial
   .
string: .db "ATmega8",0 ;строка, хранящаяся во FLASH-памяти 
 
   .eseg          ;начало сегмента EEPROM-памяти
_var:  .byte 2    ;резервирование 2-ух байт под переменную _var
_cnst: .db   0xAA ;резервирование байта под переменную _cnst = 0xAA


Директива .org позволяет задать компилятору начальный адрес в пределах сегментов кода, данных и EEPROM-памяти. В случае применения в сегменте кода, директива определяет адрес размещения 16-разрядного слова программ.
Директива .org
Синтаксис написания:
.org {начальный адрес}
Пример использования:
	1
2
3
4
5
6
7
8
9
10
11
12
13
14
15
16
17
18
	.equ    SRAM_START = 0x60
.equ    RAMEND     = 0x045F
 
   .dseg          ;начало сегмента данных
   .org SRAM_START ;резервирование 32 байт в SRAM под буфер,
buffer: . byte 32 ;начиная с адреса 0x60
 
   .cseg         ;начало сегмента кода
   .org 0         ;вектор сброса по адресу 0
   rjmp  initial
   .
   .org 0x50      ;начало основной программы с адреса 0x50
initial:           
   ldi   temp,high(RAMEND) ;инициализация стека 
   out   SPH,temp
   ldi   temp,low(RAMEND)
   out   SPL,temp
   .


Директивы .macro, .endmacro (.endm), определяющие начало и конец макроса соответственно.
Директивы .macro, .endmacro (.endm)
Синтаксис написания:
.macro {имя макроса}
Пример использования:
	1
2
3
4
5
6
	.macro set_bit ;объявление макроса установки бита порта 
sbi  @0,@1   ;установить бит @1 регистра порта @0
sbi  @0-1,@1 ;настроить на вывод линию @1 регистра DDRx    
.endm
.
set_bit PORTB,0 ;установить на линии 0 порта B лог.1


Директива .listmac разрешает расширенный вывод текста макросов в файле листинга. В этом случае содержимое каждого макроопределения, встретившегося в программе, отображается целиком. Если директива не используется, то код в нутрии макроса не приводится.
Директива .listmac
Синтаксис написания:
.listmac
Пример использования:
	1
	.listmac ;разрешить разворачивать текст макросов в файле листинга


Директивы .message, .warning, .error предназначены для вывода в окно сборки проекта дополнительной информации о ходе компиляции программы. Директива .message генерирует сообщение для строки, в которой был встречен ее вызов. Применение .warning приводит к выдачи предупреждения, а .error – к сообщению об ошибки. В последнем случае сборка проекта прекращается.
Директивы .message, .warning, .error
Синтаксис написания:
.message "{текст сообщение}"
.warning "{текст предупреждения}"
.error "{текст соодщения об ошибки}"
Пример использования:
	1
2
3
	.message "Macros has been called here."
.warning "Too high frequency!"
.error "Wrong macro argument!"


Группа директив условной компиляции .ifdef, .ifndef, .if, .else, elif, .endif используются для вставок программного кода в зависимости от различных условий. Директива .ifdef проверяют наличие объявления некоторого символьного имени. За директивой может следовать набор команд, которые будут подставлены в текст, если условие проверки “истина” (имя было объявлено). Директива .ifndef противоположна .ifdef проверяет отсутствие объявления символьного имени. Директива .if производит подстановку кода, когда выполняется условие сравнения, указанное в качестве ее параметра. Команды, которые должны выполняться, в случае если условие директивы .if “ложно” – располагаются после директивы .else. Ветвление типа “если” - “то” может иметь несколько уровней вложения благодаря директиве .elif. Каждый блок проверки, начинающийся с .ifdef, .ifndef, .if, должен быть закрыт директивой .endif.
Директивы if, .ifdef, .ifndef, .else, elif, .endif
Синтаксис написания:
.ifdef {символ} (или .ifndef {символ})
.if {условие}
.else {выражение} (или .elif { условие}) 
.endif
Пример использования:
	1
2
3
4
5
6
7
8
9
10
11
12
13
14
15
16
17
18
19
20
21
22
23
	.macro del_ms ;макрос, формирующий задержку времени в мс
 .ifndef FREQ ;если не объявлена константа FREQ (частота в Гц),
  .warning "Undefined FREQ constan!" ;выдаем предупреждение и
  .equ FREQ = 1000000 ;присваиваем по умолчание значение 1 МГц
 .endif
 .equ DELAY = (@0*FREQ)/4000 ;величина задания задержки времени
 .if DELAY > 65535           ;если DELAY размером больше 2 байт, то
  .error “Integer overflow in DELAY!” ;реализация макроса не возможна
 .else                              
  push XL   ;сохраняем в стеке рабочие регистры XL, XH
  push XH
  ldi  XH,high(DELAY) ;цикл задержки времени
  ldi  XL,low(DELAY)
  sbiw XH:XL,1
  brne PC-1
  pop  XH
  pop  XL   ;восстанавливаем из стека рабочие регистры XH, XL
 .endif
.endm
 .
 .equ FREQ = 2000000 ;объявление тактовой частоты 2 МГц
 .
del_ms 25          ;формирование задержки времени в 25 мс



В начале любой программы, как правило, подключается один из стандартных заголовочных файлов, поставляемых вместе с пакетом программ AVR Studio.
Заголовочные файлы существуют для всех моделей AVR. По умолчанию исходной директорией их размещения является 
С:\ProgramFiles\Atmel\AVRTools\AvrAssembler2\Appnotes”. 
Каждый файл имеет имя подобное "m8def.inc", в котором первая часть указывает на модель микроконтроллера. Заголовочный файл содержит объявление всех символьных имен, которые могут быть использованы в программе. В первую очередь к ним относятся названия управляющих РВВ и их битовых полей:
	1
2
3
4
5
6
7
8
9
10
11
12
13
14
15
16
17
18
19
20
21
22
23
24
25
26
	; ***** I/O REGISTER DEFINITIONS **************************
; NOTE:
; Definitions marked "MEMORY MAPPED" are extended I/O ports
; and cannot be used with IN/OUT instructions
 
.equ    SREG    = 0x3F
.equ    SPL = 0x3D
.equ    SPH = 0x3E
.
.equ    TWAR    = 0x02
.equ    TWSR    = 0x01
.equ    TWBR    = 0x00
 
; ***** BIT DEFINITIONS ***********************************
.
; ***** CPU **************************
; SREG - Status Register
.equ    SREG_C = 0 ; Carry Flag
.equ    SREG_Z = 1 ; Zero Flag
.equ    SREG_N = 2 ; Negative Flag
.equ    SREG_V = 3 ; Two's Complement Overflow Flag
.equ    SREG_S = 4 ; Sign Bit
.equ    SREG_H = 5 ; Half Carry Flag
.equ    SREG_T = 6 ; Bit Copy Storage
.equ    SREG_I = 7 ; Global Interrupt Enable
.


Директива .equ связывает название каждого из РВВ с его фактическим адресом в адресном пространстве ввода-вывода. Эти адреса могут использоваться только совместно с инструкциями in, out, sbi, cbi, sbis, sbic. Если обращаться к управляющему регистру как к ячейке памяти в абсолютном адресном пространстве, то к его имени необходимо будет добавлять смещение 0x20:
  lds SPL+0x20,R16 ;копировать R16 в ячейку 0x5D SRAM (0x3D I/O).
Обозначения РВВ и их битов в точности совпадают с теми, которые приводятся в технической документации на каждую конкретную модель. Поэтому не рекомендуется изменять эти имена в заголовочном файле по своему усмотрению. Стандартные названия помогают облегчить переносимость программ в пределах одного семейства. Если, например, понадобиться перенести рабочий код с модели ATmega8 на ATmeg8535, то все что надо будет сделать – это заменить в исходном тексте файле "m8def.inc" на "m8535def.inc" и заново откомпилировать проект.
В каждом заголовочном файле РОНам R27…R31 директивой .def присвоены дополнительные служебные имена:
	1
2
3
4
5
6
7
	; ***** CPU REGISTER DEFINITIONS **************************
.def    XH = R27
.def    XL = R26
.def    YH = R29
.def    YL = R28
.def    ZH = R31
.def    ZL = R30


Файл содержит также объявление ряда констант, которые часто встречаются в процессе программирования (размеры различных областей памяти, начальные адреса секции загрузчика и т.д.):
	1
2
3
4
5
6
7
8
9
10
11
12
13
14
15
16
17
18
19
20
21
22
23
	; ***** DATA MEMORY DECLARATIONS **************************
.equ    FLASHEND     = 0x0FFF  ; Note: Word address
.equ    IOEND          = 0x003F
.equ    SRAM_START   = 0x0060
.equ    SRAM_SIZE    = 1024
.equ    RAMEND   = 0x045F
.equ    XRAMEND  = 0x0000
.equ    E2END          = 0x01FF
.equ    EEPROMEND    = 0x01FF
.equ    EEADRBITS    = 9
̣̣̣̣̣̣̣̣
; ***** BOOTLOADER DECLARATIONS ***************************
.equ    NRWW_START_ADDR = 0xC00
.equ    NRWW_STOP_ADDR  = 0xFFF
.equ    RWW_START_ADDR  = 0x0
.equ    RWW_STOP_ADDR   = 0xBFF
.equ    PAGESIZE        = 32
.equ    FIRSTBOOTSTART  = 0xF80
.equ    SECONDBOOTSTART = 0xF00
.equ    THIRDBOOTSTART  = 0xE00
.equ    FOURTHBOOTSTART = 0xC00
.equ    SMALLBOOTSTART  = FIRSTBOOTSTART
.equ    LARGEBOOTSTART  = FOURTHBOOTSTART


В конце файла приведена таблица векторов прерывания:
	1
2
3
4
5
6
7
8
9
10
	; ***** INTERRUPT VECTORS *********************************
.equ    INT0addr = 0x0001  ; External Interrupt Request 0
.equ    INT1addr = 0x0002  ; External Interrupt Request 1
.equ    OC2addr  = 0x0003  ; Timer/Counter2 Compare Match
̣̣̣̣̣̣̣̣
.equ    ACIaddr  = 0x0010  ; Analog Comparator
.equ    TWIaddr  = 0x0011  ; 2-wire Serial Interface
.equ    SPMRaddr = 0x0012  ; Store Program Memory Ready
 
.equ    INT_VECTORS_SIZE = 19 ; size in words



Структура программы
Рассмотрим синтаксические особенности ассемблера AVR на примере небольшой тестовой программы для микроконтроллера ATmega8, приведенной ниже. Программа формирует на выводе PB2 импульсы с частотой следования ≈ 2.5 Гц при частоте внутреннего RC-генератора 1 МГц.
	1. ;  Тестовая программа для ATmega8.
2. // Светодиод подключен к выводу PB0 микроконтроллера.        
3. /* Биты конфигурации:  
        Low Fuse                     High Fuse     
      BODLEVEL = 1                  RSTDISBL = 1      
      BODEN    = 1                  WDTON    = 1
      SUT1     = 1                  SPIEN    = 1   
      SUT0     = 0                  CKOPT    = 1   
      CKSEL3   = 0 |                EESAVE   = 1   
      CKSEL2   = 0 |_ RC-генератор  BOOTSZ1  = 1         
      CKSEL1   = 0 |     1 МГц      BOOTSZ0  = 1      
      CKSEL0   = 1 |                BOOTRST  = 1   */ 
     
4.        .nolist ;подключение стандартного заголовочного файла
5.        .include "m8def.inc"
6.        .list 
                      
7.        .equ PAUSE  = 50000 ;задержка времени
8.        .equ LED    = PB2  ;вывод для подключения светодиода
             
9.        .def temp   = R16 ;регистр для промежуточных операций
10.       .def buffer = R17 ;регистр для чтения порта
 
11.              .cseg
12.             .org 0
13.          rjmp  initial    ;0xC01F
14.          rjmp  0 ;rjmp  service_INT0 ;внешнее прерывание 0      0xCFFE
15.          rjmp  0 ;rjmp  service_INT1 ;внешнее прерывание 1     0xCFFD
16.          rjmp  0 ;rjmp  service_OC2  ;совпадение TCNT2 и OCR2   0xCFFC
17.          rjmp  0 ;rjmp  service_OVF2 ;переполнение TCNT2        0xCFFB
18.          rjmp  0 ;rjmp  service_ICP1 ;захват в ICP1           0xCFFA
19.          rjmp  0 ;rjmp  service_OC1A ;совпадение TCNT1 и OCR1A  0xCFF9
20.          rjmp  0 ;rjmp  service_OC1B ;совпадение TCNT1 и OCR1B  0xCFF8
21.          rjmp  0 ;rjmp  service_OVF1 ;переполнение TCNT1       0xCFF7
22.          rjmp  0 ;rjmp  service_OVF0 ;переполнение TCNT0         0xCFF6
23.          rjmp  0 ;rjmp  service_SPI  ;прерывание от модуля SPI  0xCFF5
24.          rjmp  0 ;rjmp  service_URXC ;получение байта по USART   0xCFF4
25.          rjmp  0 ;rjmp  service_UDRE ;опустошение UDR в USART    0xCFF3
26.          rjmp  0 ;rjmp  service_UTXC ;передача байта по USART    0xCFF2
27.          rjmp  0 ;rjmp  service_ADCC ;прерывание от АЦП          0xCFF1
28.          rjmp  0 ;rjmp  service_ERDY ;завершение записи в EEPROM 0xCFF0
29.          rjmp  0 ;rjmp  service_ACI  ;прерывание от компаратора  0xCFEF
30.          rjmp  0 ;rjmp  service_TWI  ;прерывание от модуля TWI   0xCFEE
31.          rjmp  0 ;rjmp  service_SPMR ;завершение выполнения spm  0xCFED
 
32.             .org 0x20
33. initial: ldi   temp,low(RAMEND)                     ;0xE50F        
34.          out   SPL,temp                                          ;0xBF0D
35.          ldi   temp,high(RAMEND)                       ;0xE004
36.          out   SPH,temp                                          ;0xBF0E
37.          cbi   PORTB,LED                                     ;0x98C2
38.          sbi   DDRB,LED                                       ;0x9ABA
39.          ldi   temp,1«LED                                       ;0xE004
 
40. main:    in    buffer,PORTB                                ;0xB318        
41.          eor   buffer,temp                                        ;0x2710
42.          out   PORTB,buffer                                   ;0xBB18
43.          rcall delay                                                  ;0xD001
44.          rjmp  main                                                 ;0xCFFB
 
45. delay:   ldi   XH,high(PAUSE)                            ;0xECB3
46.          ldi   XL,low(PAUSE)                            ;0xE5A0
47.          sbiw  XH:XL,1                                       ;0x9711
48.          brne  PC-1                                              ;0xF7F1
49.          ret                                                     


В строках 1…3 приведено 3 возможных варианта оформления комментариев. Комментарий в строке 1, начинающийся со знака “;”, распознается любым ассемблером и поэтому является наиболее предпочтительным. Комментарии в строках 2 и 3 подобны тем, которые используются в нотациях языков высокого уровня. Последний из них дает возможность выделить сразу несколько строк (маркерами начала и конца фрагмента текста является “/*” и “*/” соответственно).
В строке 5 директивой .include к программе подключается стандартный заголовочный файл "m8def.inc". Необязательные директивы .nolist и .list (строки 4 и 6 соответственно) запрещают вывод содержимого подключаемого файла в файл листинга.
Все директивы допускается размещать в одной строке. Так, например, можно было бы записать
.nolist .include "m8def.inc" .list
Несмотря на это, желательно придерживаться правильного стиля программирования, в соответствии с которым в одной строке должна быть расположена только одна директива ассемблера.
Объявление констант находится в строках 7 и 8. Константе LED присваивается номер линии ввода-вывода PB2 = 2, описание которой находится в заголовочном файле. В строках 9 и 10 двум рабочим РОНам назначаются пользовательские имена.
Секция рабочего кода открывается директивой .cseg в строке 11. Директива .org 0 (строка 12) устанавливает начальный адрес в памяти программ. В строке 13 должна находиться инструкция перехода на метку начала основной программы initial.
Метка представляет собой адрес в пределах секции кода или данных. В ассемблере AVR она должна быть записана в начале строки и завершаться в конце двоеточием.
Конечно, адрес в команде может быть указан и явно. В данном примере его можно задать безошибочно (rjmp 0x20 вместо rjmp initial). Но, это возможно только потому, что директива .org 0x20 (строка 32) заставляет компилятор поместить команду в строке 33 по адресу 32-го слова FLASH-памяти программ. Однако, в большинстве случаев, адрес размещения той или иной команды заранее неизвестен и, кроме того, он может изменяться по мере того, как в программу будут вноситься изменения. Именно поэтому предпочтительней использовать метки. Назначение их адресов производится автоматически на этапе компиляции. Да и символьные имена меток, предоставляют об объектах намного больше информации, чем просо какие-то числа.
На месте каждого неиспользуемого в программе прерывания желательно поставить “заглушки” в виде команды возврата rjmp 0 или reti, как это сделано в строках 14…31.
В строках 33…39 производится инициализация ресурсов микроконтроллера. Любая программа обязательно должна начинаться с установки начального значения указателя стека (если он имеется). В строках 33…36 в SPH:SPL заносится значение RAMEND = 0x045F (вершина стека перемещается в самый верх SRAM). Для выделения младшего и старшего полубайтов константы RAMEND используются встроенные функции low(RAMEND)=0x5F и high(RAMEND)=0x04 соответственно.   
Далее разряд LED в регистре данных порта B сбрасывается на 0 (стока 37), а сама линия управления светодиодом настраивается на вывод (стока 38). В строке 39 в регистр temp заносится константа 1
	1
2
	ldi temp,(1«PB0)|(1«PB5) или ldi temp,(1«PB0)+(1«PB5)   ;temp <- 0b00100001  
out DDRB,temp                                           ;DDRB <- temp  


Код в строках 40…44 представляет собой тело основной программы. В регистр buffer считывается текущее состояние порта данных PORTB (строка 40). Далее между содержимым buffer и temp производится операция “Исключающее ИЛИ” (строка 41), после чего модифицированное содержимое buffer снова выводится в порт (строках 42). В результате такого действия логический уровень на линии PB2 изменится на противоположный. В строке 43 происходит вызов подпрограммы задержки времени delay (≈200 мс), а в строке 44 расположена инструкция перехода на метку main. Так образуется основной цикл программы, в котором происходит постоянное повторение операторов в строках 40…44: инвертирование уровня на выводе LED, задержка, затем снова инвертирование и т.д.
Внутри подпрограммы delay в регистровую пару XH:XL заносится число PAUSE = 50000, определяющее длительность задержки времени (строки 45, 46). В строке 47 из содержимого XH:XL вычитается 1. Команда условного перехода в строке 48 проверяет, значение флага Z из SREG. Если Z = 1 (результат предыдущей операции не равен нулю), то управление передается на строку 47. Так будет повторяться до тех пор, пока не выполнится условие XH:XL = 0 (т.е. пока не пройдет 50000 циклов вычитания). Команда выхода ret в строке 49 возвращает управление в то место основной программы, с которого произошел вызов delay.
Отдельно стоит обратить внимание на встроенную в ассемблер переменную PC в строке 48. Она представляет собой текущее содержимое программного счетчика. Адрес PC-1 будет указывать на предыдущее слово в памяти программ микроконтроллера. В данном случае в этом слове находится команда sbiw XH:XL,1. Использовать PC очень удобно для программных переходов в небольших пределах. Однако здесь всегда необходимо помнить, что у микроконтроллеров AVR имеются команды, которые имеют размер в 2 16-разрядных слова программ (lds Rd,k, jmp k и др.), из-за чего при вычислении смещения необходимо будет добавлять\отнимать к PC значение 2 вместо 1 на каждую такую инструкцию.
Справа, в листинге приведен машинный код, который будет сгенерирован после компиляции программы.
Создание и отладка программ
Разработка программ в AVR Studio начинается с создания проекта. После установки программы это легче всего сделать, через менеджер создания проектов во вкладке Project/Project Wizard.

[image: Project Wizard AVR Studio]
Рис.1 Project Wizard AVR Studio

После нажатия экранной кнопки New Project появится окно на рис.1, в котором надо задать название и директорию размещения проекта, например FirstProject. В качестве типа проекта необходимо выбрать Atmel AVR Assembler (проекты AVR GCC используют Си-компилятор WinAVR). Сейчас и в дальнейшем очень важно размещать все файлы проекта в одной папке, что избавит от многих проблем при редактировании и переносе программ.

[image: Project Wizard AVR Studio]

Рис.2 Project Wizard AVR Studio

На следующем этапе необходимо выбрать модель микроконтроллера и тип отладочного средства, как показано на рис.2 (в нашем случае это ATmega8 и AVR Simulator соответственно). После чего откроется окно текстового редактора, где и будет непосредственно происходить создание программы (см рис.3).
[image: Текстовый редактор AVR Studio]
Рис.3 Текстовый редактор AVR Studio

В окне проекта Project можно видеть все компоненты программы: файл с исходным текстом FirstProject.asm, заголовочный файл m8def.inc, а также выходные файлы .lst, .map, .hex и .obj с одноименными названиями. В разделе Labels находятся символьные имена меток, встречающиеся в программе.
Компиляция проекта осуществляется после нажатия на иконку Assemble либо Assemble and Run. В последнем случае сразу же запускается и программа отладчика. Если в исходным тексте были допущены ошибки, то .hex, естественно, создан не будет, а в окне Build, появится описание всех ошибок и строки где они находятся. После внесения необходимых исправлений и успешной сборки, в окне Build отобразится статистика о проекте в виде диапазонов адресов и размеров секций FLASH, SRAM и EEPROM.
[image: Работа в симуляторе AVR Studio]
Рис.4 Работа в симуляторе AVR Studio

Проверить работоспособность программы можно в симуляторе, либо с помощью любого другого отладчика. Его запуск происходит после нажатия иконки Start Debagging. На рис.4 показана работа в симуляторе и вид основных отладочных окон, в которых можно наблюдать за состоянием содержимого различных областей памяти и символьными именами, объявленных в тексте программы. Все эти данные доступны для редактирования на ходу.
Отладку можно вести как в пошаговом (кнопки Step Into, Step Over, Step Out), так в автоматическом (Auto Step) или ускоренном (Run) режимах. Имеется возможность использовать также точки останова. Симулятор, встречая строку, в которой находится точка останова, принудительно останавливает свое выполнение, после чего можно детально изучить содержимое отладочных окон. Управление точками останова производится кнопками Toggle Brekpoint и Remove all Program Brekpoints. В окне Disassembler можно видеть соответствующие машинных кодов командам ассемблера AVR.
AVR Studio предоставляет программисту интуитивно-понятный пользовательский интерфейс и при ее освоении никаких проблем, как правило, не возникает. Управление свойствами встроенного текстового редактора и среды в целом ничем не отличается от подобных действий в различных офисных приложениях. Детальное описание всех компонентов AVR Studio можно найти во встроенной справочной системе.
Структура HEX-файла
После успешной сборки проекта будет сгенерирован 16-тиричный файл FirstProject.hex, содержащий в символьном виде машинный код программы:
	1
2
3
4
5
6
7
8
	:020000020000FC
:100000001FC0FECFFDCFFCCFFBCFFACFF9CFF8CF8B
:10001000F7CFF6CFF5CFF4CFF3CFF2CFF1CFF0CFCC
:06002000EFCFEECFEDCFA3
:100040000FE50DBF04E00EBFC298BA9A04E018B3E2
:10005000102718BB01D0FBCFB3ECA0E51197F1F747
:02006000089501
:00000001FF                          


Рассмотрим формат данных hex-файл на примере второй строки текста (маркером начала строки всегда является символ “:”):
	1
2
	  NN AAAA CC DDDDDDDDDDDDDDDDDDDDDDDDDDDDDDDD SS
: 10 0000 00 1FC0FECFFDCFFCCFFBCFFACFF9CFF8CF 8B                                


Поле NN определяет количество байтов данных в строке (в нашем случае 16 байтов). Поле AAAA – это начальный адрес, с которого данные будут записаны в память микроконтроллера.
За адресом следует поле команды CC. Программатор, ориентируясь на поле CC, распознает функциональное назначения строки. Ассемблер и другие компиляторы языков высокого уровня для AVR могут установить следующие значения данного параметра:
00 – в строке находятся данные для записи в память,
01 – последняя строка в файле,
02 – строка содержит начальный адрес сегмента памяти,
04 – строка содержит адрес в пределах сегмента памяти.
В данной строке CC=00 (т.е. строка предназначена для записи данных). За полем CC (кроме команды 01) идут непосредственно данные в количестве, определяемом параметром NN. Последнее поле SS – контрольная сумма. Сумма всех байтов в неповрежденной строке без учета переполнения всегда нулевая:
   10 + 00 + 00 + 00 + 1F + … + CF + F8 + CF + 8B = 0.
Каждый hex-файл должен завершаться строкой :00000001FF.

Задание на работу.
1. Познакомиться с AVR Studio.
2. Изучить предложенные примеры.
3. Изучить интерфейс AVR Studio.
4. Изучить справочную систему AVR Studio.
5. Подготовить отчет по работе. Отчет должен содержать (тему, описание основных элементов интерфейса AVR Studio) 

Лабораторная работа 2
ПРОГРАММИРОВАНИЕ МАТЕМАТИЧЕСКИХ ОПЕРАЦИЙ В AVR STUDIO (1)
Цель работы: изучить выполнение арифметических операций над большими беззнаковыми целыми числами в приложениях для AVR-микроконтроллеров.
Краткие теоретические сведения
Рассмотрим выполнение арифметических операций над большими беззнаковыми целыми числами. В приложениях на AVR-микроконтроллерах наиболее часто приходится использовать 16-разрядные вычисления, которые достаточно легко программируются. Двухбайтовые числа предоставляют достаточно широкий диапазон представления переменных. Однако, встречаются задачи в которых необходимо применение чисел и с большей разрядностью (счётчики импульсов, накопители суммы, промежуточные результаты вычислений и т.д.).
Сложение. Реализовать многобайтовое сложение очень просто. Для этого имеется специальная команда adc Rd, Rr, которая складывает содержимое двух регистров и добавляет к полученной сумме бит переноса C из регистре SREG (Rd <- Rd+Rr+С). Этот бит устанавливается всегда, когда в результате предыдущей операции сложения возникает переполнение (т.е. бит C всегда является 9-ым битом результата операции сложения). Так может выглядеть сложение двух 16-разрядных чисел R17:R16 и R19:R18 (сумма размещается на месте второго слагаемого R19:R18):
	1
2
	add   R18,R16 ;R18 <- R18 + R16  
adc   R19,R17 ;R19 <- R19 + R17 + C 


Необходимо помнить, что в результате сложения двух n-разрядных чисел возможно образование n+1-разрядной суммы. Например, в результате следующей операции сложения получим: 0xB2FF + 0xCC45 = 0x17F44 = 0x10000 + 0x7F44.
Сумма двухбайтовых слагаемых превысила максимальное 16-разрядное значение 0xFFFF = 65535 и вместо 0x17F44 = 98116 мы получили 0x7F44 = 32580. При этом должен установиться флаг C (17-тый разряд суммы), как признак того, что произошел перенос в старший разряд и к полученному результату необходимо добавить 0x10000 = 65536.
В регистре SREG имеется еще один бит непосредственно связанный с действием сложения. Это флаг половинного переноса H, который может использоваться в 4-разрядных вычислениях. Он носит тот же смысл, что и флаг C, но указывает на переполнение суммы младших полубайтов (т.е. перенос из третьего в четвертый разряды числа). Флаг H почти никогда не используется на практике.
Если для хранения результата вычисления не хватает РОНов, то сложение рационально производить с помощью косвенной адресации, а результат размещать в SRAM процессора. В этом случае разрядность слагаемых и вычисленной суммы будет ограничена только свободным местом в памяти данных.
Подпрограмма такого сложения:
	1
2
3
4
5
6
7
8
9
10
11
12
13
14
15
16

19

22
23
24
25
	; [YH:YL] = [YH:YL] + [XH:XL] 
; [YH:YL] – первое слагаемое при входе и сумма при
;   выходе (косвенно адресуется через YH:YL)
; [XH:XL] – второе слагаемое (косвенно адресуется через XH:XL)
; R16,R17,R18 – вспомогательные регистры
; composed1 – адрес 1-го слагаемого и
; полученной суммы в ОЗУ
; composed 2 – адрес 2-го слагаемого в ОЗУ 
; SIZE – размер слагаемых в байтах 
; на выходе в C находится старший разряд результата
 
add_indirect:
     ldi   YH,high(composed1) ;заносим в указатель Y адрес
     ldi   YL,low(composed1) ;первого слагаемого composed1
     ldi   XH,high(composed2) ;заносим в указатель Y адрес
     ldi   XL,low(composed2) ;второго слагаемого composed2 
     ldi   R16,SIZE
     clc          ;при первом входе в цикл C=0      
ad1: ld    R17,X+         
     ld    R18,Y     ;поочерёдно складываем с учётом переноса 
     adc   R18,R17   ;все байты слагаемых и заносим результат
     st    Y+,R18    ;по адресу первого слагаемого
     dec   R16        
     brne  ad1       ;повторяем сложение SIZE раз    
     ret                           



Вычитание. Подобно сложению с переносом в архитектуре AVR существует и команда вычитания с заемом sbc Rd,Rr. Для связи байтов в ней тоже участвует флаг C, который в этом случае обычно называется флагом заема. Бит C устанавливается каждый раз, когда результат предыдущей операции вычитания оказывается меньше нуля и автоматически вычитается из разности полученной после команды sbc Rd,Rr (Rd<‑Rd‑Rr‑С). Ниже показан пример вычитания 16-разрядного числа R17:R16 из R19:R18 (разность помещается на место вычитаемого):
	1
2
	sub   R18,R16 ;R18 <- R18 - R16      
sbc   R19,R17 ;R19 <- R19 - R17 - C                              


Разрядность разности при вычитании никогда не превысит разрядности делимого. Однако здесь возникает другая проблема, связанная с тем, что уменьшаемое может оказаться меньше вычитаемого. В результате такого действия мы получим установленный флаг C, как признак отрицательного результата. И хотя с точки зрения арифметики такая операция является вполне законной - она приводит к отрицательным числам, представленным в дополнительном коде, которые будут рассмотрены в следующем разделе.
Ниже приведена подпрограмма вычитания двух многобайтовых чисел, размещенных в SRAM.

	; [YH:YL] = [YH:YL] - [XH:XL]
; [YH:YL] – уменьшаемое при входе и разность при 
;   выходе (косвенно адресуется через YH:YL)
; [XH:XL] – вычитаемое (косвенно адресуется через XH:XL)
; R16,R17,R18 – вспомогательные регистры
; reduced - адрес уменьшаемого и
; полученной разности в ОЗУ 
; subtracted - адрес вычитаемого в ОЗУ
; SIZE – размер уменьшаемого и вычитаемого в байтах
; на выходе бит C=1 если [YH:YL] < [XH:XL]
  
sub_indirect:       
     ldi   YH, high(reduced)        ;заносим в указатель Y адрес
     ldi   YL, low(reduced)    ;уменьшаемого reduced
     ldi   XH, high(subtracted) ;заносим в указатель X адрес
     ldi   XL, low(subtracted) ;вычитаемого subtracted
     ldi   R16,SIZE
     clc          ;при первом входе в цикл C=0      
sb1: ld    R17,X+         
     ld    R18,Y     ;поочерёдно вычитаем с учётом заёма из
     sbc   R18,R17   ;уменьшаемого все байты вычитаемого и  
     st    Y+,R17    ;заносим результат по адресу уменьшаемого
     dec   R16        
     brne  sb1       ;повторяем вычитание SIZE раз 
     ret                             



Умножение. У микроконтроллеров AVR, по сути, имеется два пути реализации умножения, разница между которыми заключается в форме представления произведения и, соответственно, в различии выполняемых арифметических операций. Первый оптимизирован для использования с командой однобайтового умножения, второй – использует только сложение и сдвиговые операции. Для произведения n-разрядного множимого и m-разрядного множителя необходимо отводить n+m разрядов.
Помимо этого существует особый случай умножения на целую степень 2. Всё дело в том, что 2 является основанием двоичной системы, а в любой позиционной системе операция умножения числа на основание системы приводит к увеличению веса каждого его разряда на порядок:
2*X = 2*(xn-1*2n-1 + xn-2*2n-2 + … + x1*21 + x0*20) = xn-1*2n + xn-2*2n-1 + … + x1*22 + x0*21.
Как видно, коэффициенты xn числа X переместились на один разряд влево (x0 стал при 21, x1 стал при 22 и т.д.). Итак, для умножения двоичного числа на 2n необходимо просто сдвинуть его на n разрядов влево. Так можно осуществить умножение на 2 двухбайтового числа в регистрах R17:R16 посредством сдвиговых операций:
	1
2
	lsl   R16 ;R16 <- R16 << 1 (LSB <- 0, флаг С <- MSB)  
rol   R17 ;R17 <- R17 << 1 (LSB <- флаг С, флаг С <- MSB)                            


Тот же результат можно получить, складывая число с самим собой:
	1
2
	add   R16, R16      
adc   R17, R17                                  


Благодаря тому, что операции сдвига и сложения выполняются за одинаковое время (1 машинный цикл) оба примера равноценны.
Иногда встречаются операции умножения на число, которое близко к степени 2. В таких случаях намного проще заменить его суммой двух слагаемых, одно из которых кратно 2n, и использовать только операции сдвига и сложения (вычитания): 
63*X = (26-1)*X = 26*X - X = X<<6 – X,
33*X = (25+1)*X = 25*X + X = X<<5 + X,
130*X = (27+2)*X = 27*X + 2*X = X<<7 + X<<2.
Для умножения 2-байтовых чисел (обозначим их как 
XH:XL = 28*XH + XL и YH:YL = 28*YH + YL) 
применяется следующая вычислительная схема:
XH:XL * YH:YL = (28*XH + XL)*(28*YH + YL) = 216*XH*YH + 28*(XH*YL + YH*XL) + XL*YL.
[image: Произведение Z двух произвольных двоичных чисел]

Отыскание 32-разрядного результата сводится к последовательности операций однобайтовых умножений и последующего сложения всех произведений с учётом сдвига слагаемых XH*YL, XL*YH на 1 байт и XH*YH на 2 байта, как показано на рис.1. Подпрограмма, реализующая эти действия, приведена ниже. Напомним, что произведение двух однобайтовых множителей, полученное в результате выполнения команды mul Rd,Rr или любой другой команды умножения, всегда заносится в регистровую пару R1:R0.

	; R23:R22:R21:R20 = R17:R16 * R19:R18
; R23:R22:R21:R20 - произведение
; R17:R16 – множимое
; R19:R18 – множитель
; R1,R0 – вспомогательные регистры
 
mul16_16:
     mul   R16,R18 ;находим XL*YL = R16*R18 и заносим его в
     movw  R20,R0  ;младшие байты произведения R21:R20
     mul   R17,R19 ;находим XH*YH = R17*R19 и заносим его в
     movw  R22,R0  ;старшие байты произведения R23:R22
     mul   R17,R18 ;находим XH*YL = R17*R18 и прибавляем его к
     clr   R17     ;байтам R23:R22:R21 произведения
     add   R21,R0   
     adc   R22,R1      
     adc   R23,R17   
     mul   R16,R19 ;находим YH*XL = R19*R16 и прибавляем его к 
     add   R21,R0  ;байтам R23:R22:R21 произведения 
     adc   R22,R1
     adc   R23,R17  
     ret                                


Возвести двухбайтовое число в квадрат еще проще:
XH:XL * XH:XL = (28*XH + XL)*(28*XH + XL) = 216*XH*XH  + 28*2*XH*XL + XL*XL.
Подпрограмма возведения в квадрат приведена ниже. Она является хорошим примером того, где может пригодиться инструкция fmul Rd,Rr, которая одним действием производит умножение двух однобайтовых чисел и сдвиг произведения на 1 разряд влево.
	1
2
3
4
5
6
7
8
9
10
11
12
13
14
15
16
17
	; R21:R20:R19:R18 = R17:R16 * R17:R16
; R17:R16 – число, возводимое в квадрат
; R21:R20:R19:R18 – произведение
; R1,R0 – вспомогательные регистры
      
sqr16_16: 
    mul   R16,R16 ;находим XL*XL = R16*R16 и заносим его в
    movw  R18,R0  ;младшие байты произведения R19:R18
    mul   R17,R17 ;находим XH*XH = R17*R17 и заносим его в
    movw  R20,R0  ;старшие байты произведения R21:R20
    fmul  R17,R16 ;находим 2*XH*YL = R17*R18 и прибавляем его 
    clr   R17     ;к байтам R21:R20:R19 произведения
    rol   R17
    add   R19,R0
    adc   R20,R1
    adc   R21,R17
    ret                                


Во многих случаях команда fmul Rd,Rr позволяет использовать  аппаратный умножитель 8x8 фактически как умножитель 9x8 с получением 17-разрядного результата (MSB произведения размещается в C). Это бывает очень полезно, когда возникает необходимость умножить переменную на постоянный числовой коэффициент, который немного выходит за пределы 8-разрядной сетки. Так, допустим, можно умножить число из R16 на 500

	ldi   R17,250 ;R1:R0 <- (250<<1)*R16 = 500*R16
fmul  R17,R16 ; флаг С <- MSB                               


или на 1000:
	1
2
3
4
	ldi   R17,250 ;R1:R0 <- (250<<2)*R16 = 1000*R16
fmul  R17,R16 ;флаг С <- MSB
lsl   R0
rol   R1                             


Ниже показан другой практически важный пример умножения, когда один из множителей (множитель X) однобайтовый. В основе подпрограммы лежит следующая вычислительная схема: 
X * YH:YL = X*(28*YH + YL) = 28*X*YH + X*YL

	; R21:R20:R19 = R18 * R17:R16
; R21:R20:R19 - произведение 
; R18 – множимое
; R17:R16 – множитель
; R1,R0 – вспомогательные регистры
 
mul8_16:
     mul   R18,R17 ;находим X*YH = R18*R17 и заносим его в   
     movw  R20,R0  ;старшие байты произведения R21:R20   
     mul   R18,R16 ;находим X*YL = R18*R16   
     clr   R18       
     add   R19,R0  ;и прибавляем его  
     add   R20,R1  ;к произведению R21:R20:R19  
     adc   R21,R18     
     ret                          


Точно также могут быть разложены числа и с большей разрядностью. Однако с ростом их величины начинают резко возрастать и затраты ресурсов процессора. Так для умножения трёхбайтовых чисел понадобится по 9 операций однобайтовых умножений и двухбайтовых сложений; для четырёхбайтовых уже по 16 операций и т.д.
В подобных случаях необходимо использовать другой, наиболее общий алгоритм, который не использует команду mul Rd,Rr. А для микроконтроллеров семейства ATtiny, (а также для устаревшей линейки моделей Classic), у которых отсутствует аппаратный умножитель, он вообще является единственно возможным.
Для пояснения алгоритма, перепишем произведение Z двух произвольных двоичных чисел
[image: Произведение Z двух произвольных двоичных чисел]
Произведение X*yj называется частичным произведением (X*yj = X при yj =1 и X*yj = 0 при yj =0), а произведение (X*yj)*2j  есть не что иное, как частичное произведение, сдвинутое на j разрядов влево. Итак, нахождение произведение X*Y сводится к нахождению суммы частичных произведений X*yj, каждое из которых в свою очередь сдвинуто на j разрядов влево соответственно. На этом принципе основан школьный метод умножения в столбик. Рассмотрим пример умножения двух 4-разрядных чисел:
[image: Умножение двух 4-разрядных чисел]
Как видим, для получения результата используются только операции сложения и сдвига. При этом в тех разрядах Y, где yj=0 и X*yj=0. И если предварительно анализировать значения yj/, то можно пропускать пустое сложение с 0 в соответствующих итерациях. Ниже приведена подпрограмма умножения двух трёхбайтовых чисел. В ней для получения суммы частичных произведений, сдвинутых каждое на j разрядов влево, производится сдвиг накопителя произведений m=24 раза вправо, а для экономии памяти младшие 3 байта произведения заносятся в те же регистры, где находился множитель.

	; R24:R23:R22:R21:R20:R19 = R18:R17:R16 * R21:R20:R19 
; R24:R23:R22:R21:R20:R19 – произведение
; R18:R17:R16 – множимое
; R21:R20:R19 – множитель
; R25,R1,R0 – вспомогательный регистр
  
mul24_24:
     clr   R22    ;очищаем регистры R22,R23,R24 
     clr   R23    ;при входе в подпрограмму
     clr   R24
     ldi   R25,24 ;инициализируем счётчик циклов сложения 
     clc          ;сдвинутых частичных произведений
ml1: sbrs  R19,0  ;если младший бит множителя 1, то
     rjmp  ml2  
     add   R22,R16 ;добавляем к накопителю очередное  
     adc   R23,R17 ;частичное произведение R18:R17:R16 
     adc   R24,R18         
ml2: ror   R24    ;байтам R24:R23:R22 накопителя произведения 
     ror   R23    ;в ином случае пропускаем это действие
     ror   R22    ;и переходим к очередному сдвигу множителя
     ror   R21     
     ror   R20
     ror   R19
     clc
     dec   R25    ;повторяем m=24 раз цикл сложения
     brne  ml1    ;сдвинутых частичных произведений
     ret                    


Задание на работу.
6. Изучить предложенные примеры.
7. Выполнить задание:
	№ варианта
	Задание

	1
	Написать программу для сложения трех 16-разрядных чисел, для умножения трех 8-разрядных чисел.   

	2
	Написать программу для сложения трех 8-разрядных чисел, для умножения двух 16-разрядных чисел.   

	3
	Написать программу для вычитания трех 16-разрядных чисел, для умножения трех 8-разрядных чисел.   

	4
	Написать программу для вычитания двух 24-разрядных чисел, для умножения трех 8-разрядных чисел.   

	5
	Написать программу для сложения двух 32-разрядных чисел, для умножения трех 8-разрядных чисел.   

	6
	Написать программу для сложения трех 16-разрядных чисел, для умножения двух 16-разрядных чисел.   

	7
	Написать программу для вычитания трех 32-разрядных чисел, для умножения трех 8-разрядных чисел.   

	8
	Написать программу для вычитания четырех 8-разрядных чисел, для умножения двух 16-разрядных чисел.   

	9
	Написать программу для вычитания трех 16-разрядных чисел, для умножения трех 8-разрядных чисел.   

	10
	Написать программу для сложения трех 16-разрядных чисел, для умножения трех 8-разрядных чисел.   

	11
	Написать программу для сложения трех 16-разрядных чисел, для умножения трех 8-разрядных чисел.   

	12
	Написать программу для вычитания трех 16-разрядных чисел, для умножения трех 8-разрядных чисел.   

	13
	Написать программу для сложения четырех 32-разрядных чисел, для умножения трех 16-разрядных чисел.   

	14
	Написать программу для вычитания двух 32-разрядных чисел, для умножения трех 8-разрядных чисел.   

	15
	Написать программу для сложения двух 16-разрядных чисел, для умножения трех 8-разрядных чисел.   

	16
	Написать программу для сложения трех 16-разрядных чисел, для умножения двух 24-разрядных чисел.   

	17
	Написать программу для вычитания трех 32-разрядных чисел, для умножения трех 8-разрядных чисел.   

	18
	Написать программу для сложения трех 16-разрядных чисел, для умножения трех 8-разрядных чисел.   

	19
	Написать программу для вычитания трех 32-разрядных чисел, для умножения трех 8-разрядных чисел.   

	20
	Написать программу для сложения четырех 16-разрядных чисел, для умножения трех 24-разрядных чисел.   



8. Составить контрольные примеры.
9. Показать результаты работы преподавателю.
10. Подготовить отчет по работе. Отчет должен содержать (тему, цель работы, тексты программ, контрольные примеры, результаты работы программ). 
11. Ответить на контрольные вопросы.

Контрольные вопросы
1. Как реализовать многобайтовое сложение в приложениях на AVR-микроконтроллерах?
2. Как реализовать многобайтовое вычитание в приложениях на AVR-микроконтроллерах?
3. Как реализовать многобайтовое умножение в приложениях на AVR-микроконтроллерах?

Лабораторная работа 3
ПРОГРАММИРОВАНИЕ ОПЕРАЦИИ ДЕЛЕНИЕ В AVR STUDIO 
Цель работы: изучить выполнение операции деление над беззнаковыми числами в приложениях для AVR-микроконтроллеров.
Краткие теоретические сведения
Деление. Из всех арифметических операций деление занимает особое место. По отношению к умножению деление является обратной операцией и не имеет конечной формулы для определения частного. Деление – единственная арифметическая операция ограниченной точности! Не смотря на это алгоритмы деления достаточно просты. Ниже будут показаны два таких алгоритма, которые используют только операции вычитания и сдвигов. Для того чтобы получить результат в виде целочисленных частного и остатка нужно предварительно убедиться в том, чтобы делимое было больше делителя и, конечно, исключить возникновение запрещённой операции деления на 0. При делении n-разрядного делимого на m-разрядный делитель под частное необходимо отвести n, а под остаток m разрядов.
Частным случаем является деление на целую степень 2:
X/2 = (x-1*2n-1 + xn-2*2n-2 + … + x1*21 + x0*20 )/2 = xn-1*2n-1 + xn-2*2n-2 + … + x1*20 + x0*1/21
Все коэффициенты xn двоичного числа X переместились на один разряд вправо (x0 стал при 1/21, x1 стал при 20 и т.д.). Таким образом для деления двоичного числа на 2nнеобходимо произвести сдвиг его содержимого на n разрядов в правую сторону. Так выглядит деление на 2 16-разрядного числа в регистрах R17:R16:

	lsr  R17 ; R17 <- R17 >> 1 (MSB <- 0, флаг С <- LSB)    
ror  R16 ; R16 <- R16 >> 1 (MSB <- С, флаг С <- LSB)                              


Обратите внимание на то, что после сдвига вправо во флаге переноса С окажется целочисленный остаток (младший коэффициент x0) от деления на 2.
Для другого частного случая деления на 3 существует один очень интересный алгоритм, основанный на разложении дроби X/3 в ряд вида:
   X/3 = X/2 - X/4 + X/8 - X/16 + X/32 - …
Каждый член ряда получается делением X на целую степень 2, что как было показано выше, легко реализуется сдвиговыми операциями. Ниже приведена подпрограмма деления на 3 16-разрядного числа.

	; R19:R18 = R17:R16/3
; R19:R18 – частное
; R17:R16 – делимое
; R20 – вспомогательный регистр
  
div16to3:
    clr   R18     ;очищаем вспомогательные регистры R18,R19
    clr   R19     ;при входе в подпрограмму   
du1: rcall shft_right
    brne  PC+2    ;если Z=1, 
    ret           ;то завершаем деление
    add   R18,R16 ;в ином случае добавляем к накопителю
    adc   R19,R17 ;очередной нечётный член ряда
    rcall shift_right
    brne  PC+2    ;если Z=1,
    ret           ;то завершаем деление
    sub   R18,R16 ;в ином случае вычитаем из накопителя
    sbc   R19,R17 ;очередной чётный член ряда
    rjmp  du1
 
shft_right: 
    lsr   R17     ;производим деление R17:R16 / 2,
    ror   R16     ;получая очередной член ряда
    mov   R20,R17 ;если R20 = R17+R16 = 0 (т.е. R17:R16=0), 
    or    R20,R16 ;то выходим из подпрограммы с флагом Z=1
    ret                            


Этот пример очень эффективен. Его быстродействие ограничено только разрядностью делимого. Более того для делимого произвольной величины обрабатывается оптимальное число членов ряда (до 16), а результат автоматически округляется до ближайшего целого числа.    
Естественно, что в тех случаях, когда необходимо разделить число на 6 можно совместно использовать приемы деления на 2 и 3:
X/6 = (X/2)/3 = (X >> 1)/3 = (X/3) >> 1
Для уменьшения погрешности деление чётных чисел следует начинать с деления на 2 (остаток 0) а, деление нечётных с деления на 3 (алгоритм этого вида деления учитывает остаток).
В общем случае самый естественный и простой способ разделить одно число на другое – это решить уравнение, вытекающее непосредственно из определения операции деления:
X = Z*Y + R,
R < Y, Y ≠ 0,
где X – делимое, Y – делитель, Z – частное, R – целочисленный остаток от деления.
Уравнение содержит два неизвестных параметра Z и R и поэтому не может быть явно разрешено. Для отыскания результата необходимо прибегнуть к итерационному методу: вычитать из делимого делитель до тех пор, пока остаток не окажется меньше делителя. Тогда число циклов вычитания численно даст частное Z, а остаток будет равен целочисленному остатку R от деления. Для примера, произведем следующее деление:
X = 213, Y = 10,
X - Z*Y = R, 213 – 21*10 = 3,
Z = 21, R = 3.    
Мы 21 раз смогли вычесть из 213 по 10 пока не образовался остаток 3<10. Приведённый алгоритм имеет существенный недостаток – скорость его выполнения напрямую зависит от величины частного (числа итераций вычитания), что делает нежелательным его использование для деления больших чисел. Применяется он, в основном, в задачах ограниченных делением однобайтовых величин. Подпрограмма деления:

	    ; [R18] + {R17} = R17 / R16 
; R18 – частное 
; R17 – делимое при входе и целочисленный остаток на выходе
; R16 – делитель
 
div8_8:   
     clr   R18      ;очищаем R18 при входе
     sub   R17,R16  ;производим вычитание R17-R16
     inc   R18         
     brcc  PC-2     ;до тех пор пока разность R17-R16 > 0
     dec   R18      ;когда разность R17-R16 < 0
     add   R17,R16  ;восстанавливаем R17 и корректируем R18
     ret          


Для чисел большей разрядности необходимо использовать способ деления, основанный на приведенной ниже вычислительной схеме. Для этого представим необходимо выражение операции деления в следующем виде:
[image: Операция деления]
Нахождение частного сводится к нахождению его коэффициентов zi, которые определяются, согласно формуле, следующим образом: необходимо последовательно сравнивать X с произведениями Y*2i и если  X > Y*2i, то в этом случае необходимо произвести вычитание Y*2i из X, а соответствующий разряд zi установить в 1; если X < Y*2i – вычитание пропускается и в этой итерации zi=0. Эта процедура продолжается до тех пор, пока не останется остаток Ri получаются простым сдвигом Y на i разрядов влево. Аналогично производится деление в любой позиционной системе (метод деления в столбик), но проще всего в двоичной из-за того, что в X может содержаться Y*2i максимум один раз (на что и указывает единица в соответствующем разряде). Рассмотрим пример деления:
[image: Пример деления]
Необходимо обратить внимание на то, что число итераций сравнения должно совпадать с числом значащих разрядов частного (в данном случае n=4 для определения z0…z3). Однако разрядность частного заранее никогда не известна и поэтому всегда необходимо знать максимально возможную разрядность результата деления. На практике чаще всего используют вычисления, в которых частное заведомо умещается в 8,16,24 бита (кратно одному байту) и т.д. При этом нужно использовать 8,16,24 итераций сравнения X с Y*2i, соответственно. С целочисленным остатком R проблем не возникает – его размер ограничен разрядностью Y (R < Y). 
Подпрограмма деления двухбайтового числа на однобайтовое приведена ниже. В ней для сравнения X с Y*2i (i ∈ {0,…,15}) вместо сдвига делителя Y на n разрядов вправо используется поочерёдный сдвиг делимого на n разрядов влево, а для экономии памяти частное записывается на тоже место, что и делимое. В начале программы осуществляется проверка условия Y ≠ 0, без которого деление не может быть корректно осуществлено.

	; [R17:R16] + {R18} = R17:R16 / R20 
; R17:R16 – делимое при входе и частное на выходе
; R20 – делитель
; R18 – целочисленный остаток
; R21, R22 – вспомогательные регистры
; при выходе из подпрограммы в C находится признак ошибки
; если C = 1 – произошла ошибка (R20 = 0)
; если С = 0 – деление успешно выполнено 
 
div16_8:  
     tst   R20        ;если R20=0 выходим из подпрограммы
     breq  dv3        ;с признаком ошибки C=1
     clr   R18        ;очищаем регистры R18,R19,R21 при  
     clr   R19        ;входе в подпрограмму  
     clr   R21
     ldi   R22,16     ;инициализируем счётчик циклов      
dv1: lsl   R16        ;сдвигаем делимое R17:R16 со    
     rol   R17        ;вспомогательными регистрами  
     rol   R18        ;R19,R18 на один разряд влево   
     rol   R19
     sub   R18,R20    ;осуществляем пробное вычитание 
     sbc   R19,R21    ;R19:R18 - R20 и если R19:R18 > R20, 
     ori   R16,0x01   ;то устанавливаем zi=1      
     brcc  dv2
     add   R18,R20    ;в ином случае восстанавливаем делимое
     adc   R19,R21
     andi  R16,0xFE    ;и устанавливаем zi=0      
dv2: dec   R22         
     brne  dv1        ;повторяем цикл n=16 раз   
     clc              ;успешно завершаем подпрограмму  
     ret              ;со флагом С=0
dv3: sec              ;выходим из-за ошибки  
     ret              ;с флагом С=1   


Ниже приведена еще одна важная подпрограмма, реализующая  деление 4-хбайтового числа на двухбайтовое с получением 16-разрядных частного и остатка. По своей структуре она аналогична предыдущей, кроме того, что при входе в подпрограмму производится проверка на переполнение результата и тем самым исключаются случаи, при которых частное может не умещаться в отведённых 2-х байтах (например, X = 0x51F356D, Y = 0x100, R = 0x6D, Z = 0x51F35 – имеет разрядность более 16 бит).
	1
2
3
4
5
6
7
8
9
10
11
12
13
14
15
16
17
18
19
20
21
22
23
24
25
26
27
28
29
30
31
32
33
34
35
36
37
38
39
40
41
42
43
44
45
	; [R17:R16] + {R21:R20} = R19:R18:R17:R16 / R23:R24
; R17:R16 – частное
; R21:R20 - целочисленный остаток
; R19:R18:R17:R16 – делимое 
; R23:R24– делитель
; R25,R26 – вспомогательный регистр
; при выходе из подпрограммы в C находится признак ошибки
; если C = 1 – произошла ошибка (R23:R24 = 0 или размер
; частного больше 2 байт)
; если С = 0 – деление успешно выполнено 
 
div32_16:  
     mov   R20,R23   ;если R20 = R23 + R24 = 0
     or    R20,R24   ;(т.е. R24:R23 = 0), то выходим из
     breq  dv3       ;подпрограммы с признаком ошибки C=1
     cp    R18,R23   ;если R19:R18 < R24:R23
     cpc   R19,R24   ;(т.е.частное больше 2-х байт), то выходим 
     brcc  dv3       ;из подпрограммы с признаком ошибки C=1
     clr   R20       ;очищаем регистры R20,R21,R22,R25 
     clr   R21       ;при входе в подпрограмму  
     clr   R22
     clr   R25
     ldi   R26,32    ;инициализируем счётчик циклов           
dv1: lsl   R16       ;сдвигаем делимое R19:R18:R17:R16 со
     rol   R17       ;вспомогательными регистрами R20, R21, R22
     rol   R18       ;на один разряд влево
     rol   R19
     rol   R20
     rol   R21
     rol   R22
     sub   R20,R23   ;осуществляем пробное вычитание 
     sbc   R21,R24   ;R22:R21:R20 - R24:R23 и если
     sbc   R22,R25   ;R22:R21:R20 > R24:R23,         
     ori   R16,0x01  ;то устанавливаем zi=1      
     brcc  dv2           
     add   R20,R23   ;в ином случае восстанавливаем делимое
     adc   R21,R24
     adc   R22,R25
     andi  R16,0xFE  ;и устанавливаем zi=0      
dv2: dec   R26       ;повторяем n=32 раза цикл
     brne  dv1       ;сравнения R20:R19:R18 с R22:R21 
     clc             ;успешно завершаем подпрограмму
     ret             ;с флагом С=0
dv3: sec             ;выходим из-за ошибки
     ret             



Формы представления частного в операциях деления
Все подпрограммы деления, приведенные в предыдущем разделе, возвращают на выходе результат в виде целочисленных частного Z и остатка R 
 X = 11, Y = 4,  
X/Y = Z + R,
11/4 = [2] + {3}.
Однако при попытке продолжить деление, по аналогии с подобным действием в десятичной системе, что иногда бывает необходимо на практике, мы получим частное, представленное в форме дробного числа с фиксированной запятой

X = 11, Y = 4,
X/Y = Z + R,
11/4 = 0b10.11
Работа с такими числами ни чем не отличается от работы с неотрицательными целыми числами. Разница состоит лишь в том – где фиксируется дробная часть результата. Так результат деления 11/4 дал целую часть числа 2 и дробную (отделённые младшие два разряда) 0.75. Теперь если вам потребуется использовать этот результат в последующих арифметических операциях, то нужно просто запомнить где находится разделяющая запятая и соответственно определить веса каких разрядов 2n, а каких 1/2m. Тоже число 0b1011 может трактоваться по-разному в зависимости от того сколько разрядов отведено под целую и дробную части

0b1.011 = 1*20 + 0*(1/21) + 1*(1/22) + 1*(1/23) = 1.375  (формат (1.3)),
0b10.11 = 1*21 + 0*20 + 1*(1/21) + 1*(1/22)  = 2.750  (формат (2.2)),
0b101.1 = 1*22 + 0*21 + 1*20 + 1*(1/21)   =  5.500  ( формат (3.1)) и т.д.

Рассмотрим как можно, например, сложить два дробных числа, предварительно приведя их к одному виду:
11/4 + 7/8 = 0b1011/0b0100 + 0b0111/0b1000 = 0b1011 >> 2 + 0b0111 >> 3 = 0b10.11 + 0b0.111 = 0b10.110 + 0b00.111 = 0b11.101 = 3.625
Дробные двоичные слагаемые 0b10.11 = 2.750 (формат (2.2)) и 0b0.111 = 0.875 (формата (1.3)) были приведены к единому формату представления (2.3) (0b10.110, 0b00.111), где целую и дробную части отделяют запятая между третьим и четвёртым разрядами. После этого мы их сложили, как два обыкновенных семизначных числа. Итак, дробные числа с фиксированной запятой – подчиняются всем тем же законам двоичной арифметики, что и целые числа.
В случае использования дробных чисел с фиксированной запятой в промежуточных вычислениях необходимо заранее ограничить разрядность дробной части результата. Это обусловлено тем, что при делении возможно образование бесконечных дробей, как допустим, 179/9 = 0b1001.100001… = 10.515625….   
Округление двоичных чисел происходит точно так же, как и в десятичной системе исчисления. Если дробный остаток от деления ≥0.5 (т.е. если старший разряд в остатке, вес которого как раз 1/21, равен 1) или целочисленный остаток R≥0,5*Y , то частное Z округляется в большую сторону. Так при делении 9/2 = 0b100.1 = 4.5 необходимо учесть дробную часть 0.5 и округлить частное до 5, а, например, при делении 9/4 = 0b10.01 = 2.25 остаток 0.25 надо отбросить.
Другим способом произвести корректировку частного можно, если добавить к нему 1/2. Тогда можно будет свободно отбросить остаток, поскольку округление в большую сторону будет происходить каждый раз, когда R≥0,5. Добавление 1/2 к частному равносильно добавлению перед делением к делимому половины делителя
(X + Y/2) / Y = X/Y + 1/2 = Z + 1/2.
Подпрограмма, приведенная ниже вносит такую поправку перед делением 15-разрядного делимого на 8-разрядный делитель.

	; R17:R16 + R18 / 2 = (2*R17:R16 + R18) / 2
; R17:R16 – делимое, подлежащее корректировке
; R18 – делитель
; R19 – вспомогательный регистр
 
cor_div:   
     clr   R19       ;обнуляем R20 
     lsl   R16       ;умножение 2*R17:R16
     rol   R17         
     add   R16,R18   ;сложение 2*R17:R16 + R18
     adc   R17,R19
     ror   R17       ;деление (2*R17:R16 + R18)/2      
     ror   R16
     adc   R16,R19   ;корректируем (2*R17:R16 + R18)/2      
     adc   R17,R19   ;если остаток ≥0,5 
     ret                 



Задание на работу.
12. Изучить предложенные примеры.
13. Выполнить задание:
	№ варианта
	Задание

	1
	Написать программу для деления 16-разрядного числа на 8-разрядное. Использовать все возможные алгоритмы. Выполнить округление, если необходимо.   

	2
	Написать программу для деления 16-разрядного числа на 8-разрядное. Использовать все возможные алгоритмы. Выполнить округление, если необходимо.   

	3
	Написать программу для деления 16-разрядного числа на 8-разрядное. Использовать все возможные алгоритмы. Выполнить округление, если необходимо.   

	4
	Написать программу для деления 16-разрядного числа на 8-разрядное. Использовать все возможные алгоритмы. Выполнить округление, если необходимо.   

	5
	Написать программу для деления 16-разрядного числа на 8-разрядное. Использовать все возможные алгоритмы. Выполнить округление, если необходимо.   

	6
	Написать программу для деления 16-разрядного числа на 8-разрядное. Использовать все возможные алгоритмы. Выполнить округление, если необходимо.   

	7
	Написать программу для деления 16-разрядного числа на 8-разрядное. Использовать все возможные алгоритмы. Выполнить округление, если необходимо.   

	8
	Написать программу для деления 16-разрядного числа на 8-разрядное. Использовать все возможные алгоритмы. Выполнить округление, если необходимо.   

	9
	Написать программу для деления 16-разрядного числа на 8-разрядное. Использовать все возможные алгоритмы. Выполнить округление, если необходимо.   

	10
	Написать программу для деления 16-разрядного числа на 8-разрядное. Использовать все возможные алгоритмы. Выполнить округление, если необходимо.   

	11
	Написать программу для деления 16-разрядного числа на 8-разрядное. Использовать все возможные алгоритмы. Выполнить округление, если необходимо.   

	12
	Написать программу для деления 16-разрядного числа на 8-разрядное. Использовать все возможные алгоритмы. Выполнить округление, если необходимо.   

	13
	Написать программу для деления 16-разрядного числа на 8-разрядное. Использовать все возможные алгоритмы. Выполнить округление, если необходимо.   



14. Составить контрольные примеры.
15. Показать результаты работы преподавателю.
16. Подготовить отчет по работе. Отчет должен содержать (тему, цель работы, тексты программ, контрольные примеры, результаты работы программ). 
17. Ответить на контрольные вопросы.

Контрольные вопросы
1. Как реализовать деление в приложениях на AVR-микроконтроллерах?
2. Поясните формы представления частного в операциях деления в приложениях на AVR-микроконтроллерах?
3. Как выполнить округление двоичных чисел? Примеры. 

Лабораторная работа 4
РЕАЛИЗАЦИЯ ВЕТВЛЕНИЙ И ЦИКЛИЧЕСКИХ ПРОГРАММ 
Цель работы: изучение принципов реализации типовых алгоритмических структур на примере ветвлений и циклических программ.
Краткие теоретические сведения
Любая процедура управления или обработки данных представляет собой совокупность некоторых алгоритмических структур, с помощью которых выполняются требуемые операции. Наиболее распространенными алгоритмическими структурами являются ветвления (branching) и циклы (loop). Ветвления используются для выполнения различных частей программы (разделения ветвей алгоритма) в зависимости от некоторых условий.
В циклах одна и та же операция выполняется над содержимым нескольких последовательно расположенных в памяти ячеек или элементов данных. Использование циклических программ целесообразно при обработке массивов, таблиц и подобных по структуре данных. Числом повторений цикла управляют счетчики, а обрабатываемый при данном проходе цикла элемент определяется с помощью индекса или указателя.
В циклической программе можно выделить четыре основных блока.
1. Блок инициализации (от лат. initium – начало), в котором производится присвоение начальных значений переменным, счетчикам, индексам и указателям. Указатели представляют собой адреса данных в памяти.
2. Блок обработки, в котором выполняются требуемые вычисления, т.е. одинаковые повторяющиеся действия над различными последовательно расположенными в памяти данными.
3. Блок управления циклом, в котором изменяются значения счетчиков и индексов (указателей) перед выполнением следующей повторяющейся операции, а также производится проверка условия выхода из цикла.
4. Заключительный блок, в котором производится сохранение полученных результатов.
Блоки 2 и 3 составляют тело цикла (loop body). Для повышения быстродействия и сокращения размера циклических программ следует разгружать тело цикла от операций, которые могут быть выполнены за его пределами.
Для организации циклических программ, а также ветвлений в программах используются команды безусловных и условных переходов. Кроме того, для построения циклов могут применяться специальные команды циклов, выполняющие несколько действий одновременно (в системе команд AVR-микроконтроллеров отсутствуют).
Команды безусловных переходов JMP, RJMP, IJMP и EIJMP передают управление по указанному в команде адресу памяти программ. Команда JMP (Jump - переход) позволяет передавать управление внутри всего объема памяти программ. Команда RJMP (Relative Jump - относительный переход) обеспечивает переход в пределах плюс-минус 2К слов (плюс-минус 4 Кбайт) относительно текущего содержимого программного счетчика. По команде IJMP (Indirect Jump – косвенный переход) выполняется косвенный переход по адресу, указанному регистром Z; максимальное смещение составляет 64 К слов (128 Кбайт). Команда EIJMP (Extended Indirect Jump – расширенный косвенный переход) обеспечивает косвенный переход по всему объему памяти программ; для расширения программного счетчика используется регистр EIND. При выполнении команд безусловных переходов в программный счетчик загружается адрес ячейки памяти программ, на которую передается управление.
Команды условных переходов передают управление по указанному адресу памяти программ в случае выполнения некоторых условий.
Команды BRхх (Branch if … – перейти, если …) выполняют переход на расстояние ‑64…+63 слова относительно текущего содержимого программного счетчика по результатам проверки разрядов регистра состояния SREG (кодов или флагов условий). Регистр состояния SREG находится в адресном пространстве регистров ввода-вывода. Коды условий (C, Z, N, V, S, H) формируются в регистре состояния при выполнении арифметических, логических команд и команд работы с битами и представляют собой признаки результата операции. Разряд С (carry – перенос) устанавливается, если при выполнении команды был перенос из старшего разряда результата. Разряд Z (zero – нуль) устанавливается, если результат выполнения команды равен нулю. Разряд N (negative – отрицательный результат) устанавливается, если старший значащий разряд результата равен 1 (показывает знак результата, если не было переполнения разрядной сетки числа со знаком). Разряд V (overflow – переполнение) устанавливается, если при выполнении команды произошло переполнение разрядной сетки числа со знаком. Разряд [image: ](sign – знак) правильно показывает знак результата при переполнении разрядной сетки числа со знаком. Разряд H (half carry – полуперенос) устанавливается, если при выполнении команды был перенос из третьего разряда результата.
Для организации ветвлений при сравнении операндов команды BRхх используются совместно с командами CP (Compare) сравнения содержимого двух РОН, CPC (Compare with Carry) сравнения с учетом признака переноса и CPI (Compare with Immediate) сравнения с константой. Команды ветвления BRхх отличаются для операндов без знака и со знаком. Числа без знака представляются прямым кодом, числа со знаком – дополнительным кодом. 
Команды условных переходов, используемые для ветвлений при сравнении операндов, сведены в табл. 1.

											Таблица 1.

К командам условных переходов также относится команда CPSE (Compare and Skip if Equal – сравнить и пропустить, если равно), которая сравнивает содержимое двух РОН и пропускает следующую за ней команду, если содержимое одинаково. Команды SBRS, SBRC, SBIS, SBIC (Skip if Bit in Register [I/O Register] is Set [Cleared] – пропустить, если разряд в регистре общего назначения [ввода-вывода] установлен [сброшен]) пропускают следующую команду в случае выполнения соответствующего условия.
При обработке массивов в циклических программах эффективно использование косвенной адресации памяти данных с предекрементом и постинкрементом, а также косвенной адресации памяти данных со смещением.
Далее приведен фрагмент программы, в которой число 100 заносится в ячейки массива из пяти байт. Для проверки условия выхода из цикла и передачи управления используется команда BRNE. Предел повторений цикла равен 5, шаг равен –1, параметр цикла (счетчик) содержится в регистре R16.

Порядок выполнения работы
1. Дополнить фрагмент программы, необходимыми директивами. Изменить число, заносимое в ячейки массива, в соответствии с заданным вариантом (табл. 2). Выполнить программу в пошаговом режиме с помощью симулятора-отладчика.
2. Произвести изменения в программе: заменить команды ADD (сложение) и SUB (вычитание) на INC (инкремент) и DEC (декремент) соответственно.

3. В последнем варианте программы исключить команду INC за счет использования косвенной адресации памяти данных с постинкрементом.
4. Изменить порядок подсчета числа проходов цикла, задав изменение параметра цикла (счетчика) не с предела, а с нуля.
5. Составить программу пересылки массива из памяти программ в память данных. Массив в памяти программ задать директивой .db; значения элементов массива взять из таблицы (массив I) в соответствии с заданным вариантом.
6. Составить программу суммирования элементов массива. Значения элементов массива задать аналогично п. 5.
7. Составить программу поэлементного сложения двух массивов. Значения элементов массивов взять из таблицы (массив I и массив II) в соответствии с заданным вариантом.
8. Составить программу слияния двух массивов, где в результате попарного сравнения двух элементов из разных массивов образуется новый массив из наибольших элементов. Значения элементов массивов взять аналогично п. 7.
Варианты заданий

Содержание отчета
Отчет должен содержать: титульный лист, цель работы; листинги трансляции программ в соответствии с заданием.

Контрольные вопросы
1. Каковы назначение и общая структура циклических программ?
2. Какие существуют команды, используемые для организации ветвлений и циклов?
3. Опишите назначение и выполнение команд безусловных переходов?
4. Назначение и выполнение команд условных переходов?
5. Опишите механизм использования кодов (флагов) условий в командах условных переходов?
Лабораторная работа 5
РАБОТА С ПОДПРОГРАММАМИ В AVR-МИКРОКОНТРОЛЛЕРАХ
Цель работы: изучение принципов организации подпрограмм и передачи параметров; освоение команд, обеспечивающих взаимодействие вызывающей программы с подпрограммой.
Краткие теоретические сведения
В большой и сложной программе можно выделить последовательности команд, выполняющие некоторые законченные функции (процедуры). Если такие последовательности команд оформить в виде отдельных модулей – подпрограмм (routine, subroutine), то в программе эти команды могут быть заменены вызовами соответствующих подпрограмм. Такое модульное (структурное) программирование дает следующие преимущества:
· ускоряется и упрощается отладка всей программы;
· подпрограммы, реализующие универсальные функции, могут использоваться при разработке других программ;
· в одной программе могут быть объединены модули, полученные после трансляции программ, написанных на разных языках программирования.
Для взаимодействия вызывающей программы с подпрограммой необходимо выполнение следующих условий:
· вызывающей программе должно быть известно положение подпрограммы в общей структуре программы;
· должны быть определены способы вызова подпрограммы и возврата из нее;
· должен быть выбран способ обмена данными между вызывающей программой и подпрограммой.
Положение подпрограммы в общей структуре программы определяется по ее имени. Имя подпрограммы на ассемблере, задаваемое символической меткой, является стартовым адресом исполняемой части подпрограммы. Вызов подпрограммы подразумевает передачу в нее управления ходом выполнения команд. Передача управления осуществляется путем загрузки в программный счетчик (РС) стартового адреса подпрограммы. При этом необходимо обеспечить сохранение адреса возврата из подпрограммы, т.е. адреса команды, следующей за командой вызова подпрограммы. Для хранения адресов возврата из подпрограмм используется стек (stack). 
Стек – специальным образом организованная последовательность ячеек памяти с дисциплиной обслуживания «последним пришел – первым вышел» (LIFO: Last-In – First-Out). При занесении в стек новых данных предыдущие данные сохраняются, но становятся временно недоступными («опускаются»). Выгружаются из стека («поднимаются», «выталкиваются») данные в обратном порядке. Стек может быть реализован как аппаратно, так и программно. Аппаратный стек (Hardware Stack) выполняется непосредственно в составе процессора в виде набора специальных регистров и, как правило, имеет небольшую глубину. Программный стек (Software Stack) организуется в оперативной памяти; глубина определяется размером и степенью использования памяти. 
Адрес очередной свободной ячейки стека содержится в специальном регистре – указателе стека SP (Stack Pointer). При записи в стек число помещается в ячейку с адресом, содержащимся в указателе стека, после чего содержимое указателя стека уменьшается на единицу (рис. 1, а). При чтении из стека производится выборка содержимого ячейки по адресу, на единицу большему содержимого указателя стека (рис. 1, б). Таким образом, при записи стека и чтении из него содержимое указателя стека изменяется. Механизм использования стека для хранения адресов возврата из подпрограмм состоит в следующем. Когда в программе встречается команда вызова подпрограммы, в ячейку памяти, определяемую указателем стека, записывается адрес возврата; значение указателя стека декрементируется; в программный счетчик загружается стартовый адрес подпрограммы.

а) 					б)
Рис. 1. Операции записи в стек (а) и чтения из стека (б): 
*** – очередная свободная ячейка стека; Addr – адрес

Далее выполняются команды подпрограммы. Когда в подпрограмме встречается команда возврата в вызывающую программу, значение указателя стека инкрементируется; сохраненный в стеке адрес возврата загружается в программный счетчик (рис. 2). Использование стека для работы с подпрограммами позволяет одной подпрограмме вызывать другую, т.е. обеспечивает возможность вложения подпрограмм. Глубина вложения подпрограмм ограничивается размером стека.

[image: ]

Рис. 2. Механизм вызова подпрограммы и возврата в вызывающую программу

В большинстве AVR-микроконтроллеров стек размещается в оперативной памяти. Указатель стека представляет собой пару 8-разрядных регистров SPH (старший байт указателя стека) и SPL (младший байт указателя стека), находящихся в адресном пространстве регистров ввода-вывода. В микроконтроллерах, размер оперативной памяти которых не превышает 256 байт, для хранения указателя стека используется только один регистр SPL (SP). Микроконтроллеры, не имеющие оперативной памяти, содержат трехуровневый аппаратный стек. 
Организуемый в оперативной памяти стек «растет» от старших адресов к младшим. Учитывая, что начальное значение указателя стека после сброса микроконтроллера равно нулю, в инициализирующей (начальной) части программы необходимо произвести его установку, если предполагается использование хотя бы одной подпрограммы. При организации стека во внутренней оперативной памяти это может быть сделано, например, следующим образом:

Команда OUT осуществляет запись содержимого регистра общего назначения в регистр ввода-вывода; RAMEND – символическое имя адреса последней ячейки внутренней оперативной памяти. Для того чтобы использовать в программе символические имена адресов
периферийных устройств AVR-микроконтроллера, необходимо при помощи директивы .include подключить файл определения адресов периферийных устройств (inc-файл). Например, для микроконтроллера ATmega8535 следует подключить файл m8535def.inc:

.include "m8535def.inc"

К подключаемому inc-файлу должен быть указан путь в поле Include Path окна AVR Assembler, вызов которого осуществляется командой AVR Assembler Setup меню Project. При использовании директивы .include нет необходимости включать в программу директиву .device, так как она содержится в соответствующем inc-файле. Избежать появления в листинге трансляции текста inc-файла позволяет директива .nolist отключения листинга. Директива .nolist используется совместно с директивой .list включения листинга (по умолчанию режим генерации листинга включен). Исключение inc-файла из листинга трансляции может быть произведено следующим образом:

Вызов подпрограммы на языке ассемблера AVR-микроконтроллеров осуществляется командами RCALL, ICALL, CALL и EICALL. Команда RCALL (Relative Call – относительный вызов) обеспечивает вызов подпрограммы, смещение начального адреса которой относительно текущего значения программного счетчика лежит в пределах ±2К слов (±4 Кбайт), с помощью относительной адресации памяти программ. Команда ICALL (Indirect Call – косвенный вызов) производит косвенный вызов подпрограммы по адресу памяти программ, содержащемуся в регистре-указателе Z; максимальное смещение начального адреса подпрограммы составляет 64 К слов (128 Кбайт). Команда CALL (Call – вызов) обеспечивает вызов подпрограммы из памяти программ размером до 4М слов (8 Мбайт) с использованием непосредственной адресации. Команда EICALL (Extended Indirect Call – расширенный косвенный вызов) позволяет выполнять косвенный вызов подпрограмм из памяти программ размером до 4М слов (8 Мбайт). При выполнении команд RCALL, ICALL, CALL и EICALL текущее значение программного счетчика заносится в стек (2 байта в микроконтроллерах с 16‑разрядным программным счетчиком или 3 байта в микроконтроллерах с 22‑разрядным программным счетчиком). Содержимое указателя стека (SPH:SPL) уменьшается соответственно на 2 или 3.






Примеры использования команд вызова подпрограмм:

[image: ]

Для возврата из подпрограмм используется команда RET. При выполнении команды RET адрес возврата загружается из стека в программный счетчик. При этом содержимое указателя стека увеличивается на 2 или 3 в зависимости от разрядности программного счетчика. Стек может также использоваться для сохранения содержимого РОН на время выполнения подпрограмм. Для сохранения в стеке и извлечения из стека содержимого РОН служат команды PUSH и POP. Команда PUSH заносит содержимое регистра в стек по адресу, хранящемуся в указателе стека, при этом значение указателя стека уменьшается на единицу (SPH:SPL=SPH:SPL–1). Команда POP выполняет обратные действия: значение указателя стека увеличивается на единицу (SPH:SPL = SPH:SPL + 1); содержимое ячейки памяти по адресу, хранящемуся в указателе стека, загружается в регистр.
При работе с подпрограммами важно обеспечить передачу параметров из вызывающей программы в подпрограмму и возврат результатов выполнения подпрограммы обратно в вызывающую программу. В ассемблере AVR-микроконтроллеров способы обмена данными между вызывающей программой и подпрограммой не формализованы. Для передачи параметров могут использоваться регистры общего назначения, ячейки оперативной памяти и стек. 
Программы с использованием подпрограмм обычно начинаются с команды относительного перехода к основной программе, в которой прежде всего производится инициализация стека (рис. 3).

Рис. 3. Пример программы с использованием подпрограммы

Передача параметров через регистры общего назначения пригодна только для небольшого числа параметров, так как число РОН ограничено и занятые под параметры регистры уже нельзя использовать в подпрограмме для участия в других вычислениях и хранения других данных. Однако это самый простой и прозрачный способ передачи параметров, обеспечивающий самый быстрый доступ к передаваемым параметрам.
Использование оперативной памяти для передачи параметров требует жесткой регламентации правил обращения к ним. Например, можно организовать в памяти массив (таблицу) для непосредственного хранения значений передаваемых параметров или их адресов;
адрес начала массива занести в РОН. Имея начальный адрес массива, вызывающая программа и подпрограмма смогут получить доступ к требуемым параметрам.
При передаче параметров через стек перед вызовом подпрограммы передаваемые параметры заносятся в стек. Следует помнить, что после выполнения команды вызова подпрограммы в стек будет добавлен адрес возврата в вызывающую программу. Задавшись значением указателя стека в качестве базового адреса и используя косвенную адресацию памяти данных со смещением, в подпрограмме можно получить доступ к содержащимся в стеке параметрам. Например, если в основной программе перед вызовом подпрограммы занести в стек значение некоторого параметра:
[bookmark: _GoBack]

то в подпрограмме можно получить к нему доступ:

(команда IN служит для считывания содержимого регистра ввода-вывода в РОН). Аналогично можно использовать стек для хранения адреса массива передаваемых параметров, расположенного в оперативной памяти данных.

Порядок выполнения работы
1. Составить программу, использующую для вызова подпрограммы команду RCALL. Выполняемую подпрограммой функцию взять из задания лабораторной работы 4. Для передачи параметров в подпрограмму использовать регистры общего назначения. Выполнить программу в пошаговом режиме, отслеживая изменение содержимого программного счетчика, указателя стека, а также занесение в стек адреса возврата из подпрограммы.
2. Выполнить задание п. 1, используя стек для передачи параметров в подпрограмму. Проследить в симуляторе занесение передаваемых параметров в стек.
3. Выполнить задание п. 2, применив для вызова подпрограммы команду ICALL.
4. Составить отчет по работе. Отчет должен содержать: титульный лист, тему, цель работы, вариант, листинги трансляции программ в соответствии с заданием.

Контрольные вопросы
1. Каковы условия взаимодействия вызывающей программы и подпрограммы?
2. Опишите принцип организации и назначение стека.
3. Каков механизм вызова подпрограмм?
4. Какие команды работы с подпрограммами на ассемблере микроконтроллеров семейства AVR вы знаете?
5. Какие существуют способы обмена данными между вызывающей программой и подпрограммой?

